TTreeView
1. Adding and Removing Nodes 

2. Node Images 

3. TTreeNode.Level 

4. Preventing Duplicates 

5. TTreeNode.Data 

6. BeginUpdate..EndUpdate 

7. ItemID - Finding Nodes 

8. Pre-Order and Post-Order recursive tree traversals 

9. Editing TreeNode.Text 

10. A more complex example 

11. Tree Sorting 

12. Linking a TreeView and a ListView 

13. Drag and Drop 

14. Saving a TreeView to disk 

Add and remove nodes
This is quick example of adding and removing nodes to/from a TTreeView. 

Start a new project, drop a TTreeView and two TButtons onto the form 





	Control
	Caption
	Name

	TTreeView
	 
	tv_eg1

	TButton
	Add
	but_Add

	TButton
	Remove
	but_Remove




Adding a node
When adding a TTreeNode you need to know where you want to add it to. In other words you must know which node, if any, is to be the new node's parent 

A root node has no parent ie TTreeNode.Parent= nil
Other nodes are 'children' of a parent node. These node's parent may be a root node or another child node. 

Most often you'd want to add a node as a child to the currently selected item. Before you do this however you must ensure that there is a selected node. To do this test if TTreeView.Selected = nil 

   if( tv_eg1.Selected = nil ) then

   begin

   end;

There are two reasons why TTreeView.Selected may be nil

1. The TTreeView is empty, there are node to be selected 

2. There are nodes, but the user has not selected one yet 

Check if the tree is empty

   if(  tv_eg1.Items.Count = 0  ) then

   begin

   end;


We now have the methods nessasary to determine the state of the TTreeView so lets add a node 

Add the Root Node 
To add an item to a TreeView you call one of the TTreeView.Items.Addxxx
functions. Such as AddFirst,  AddFirstChild,  AddChild etc.

   with tv_eg1.Items.AddFirst(  nil,  'Root'  ) do

   begin

      Selected := true;

   end;

Since we'r adding a root node use AddFirst.  This adds a new node as a sibling

of the node passed in parameter 1.

However we'r adding the root, so there are no sibling nodes. So pass nil as

parameter 1. 

Adding a child node 
      {Get a name for the new node}
   InputQuery(  'New Node',  'Caption ?',  sText  );
      {Add the node as a child of the selected node}
   with tv_eg1.Items.AddChild(  tv_eg1.Selected,  sText  ) do

   begin

      MakeVisible;

   end;
Here the user is asked for a name for the node. Then the node is added as a child

to the currently selected node.


When you add a node to a TTreeView that node migh not be shown so the

MakeVisible command shows the node that has just been added. 



Now the full source code for but_Add's OnClick event

procedure TForm1.but_AddClick(Sender: TObject);

var

   sText : string;

begin

      {If nothing is selected}
   if(  tv_eg1.Selected = nil  ) then

   begin

         {If there is no root node}
      if(  tv_eg1.Items.Count = 0  ) then

      begin

            {Add the root node}
         with tv_eg1.Items.AddFirst(  nil,  'Root'  ) do



 begin

               {Select the root node} 

            Selected := true;

         end;

      end

      else begin

            {There is a root, so user must first select

            a node before adding a new node}
         MessageBeep(  -1  );

         ShowMessage(  'Select a parent node'  );

         Exit;

      end;

   end

   else begin

         {Get a name for the new node}
      InputQuery(  'New Node',  'Caption ?',  sText  );

         {Add the node as a child of the selected node}
      with tv_eg1.Items.AddChildFirst(  tv_eg1.Selected,  sText  ) do

      begin

         MakeVisible;

      end;

   end;

end;



Removing a node 
procedure TForm1.but_RemoveClick(Sender: TObject);

begin

      {Make sure somthing is selected, before trying to

        delete it}
   if(  tv_eg1.Selected = nil  ) then

   begin

      MessageBeep(  -1  );

      ShowMessage(  'Nothing selected'  );

      Exit;

   end;

      {Dont allow user to delete the root node}
   if(  tv_eg1.Selected.Level = 0  ) then

   begin

      MessageBeep(  -1  );

      ShowMessage(  'Cant delete the root node'  );

      Exit;

   end;

      {Delete the node}
   tv_eg1.Selected.Delete;

end;

Nothing complex here...  

1. Check that there is a selected node

2. Don't allow the root to be deleted

3. Delete the delected node

You can remove the if block that does the checking for the root, if you wish.

Node Images
Most application's user interfaces can be greatly improved by using images in the TTreeView. Nodes can have different images, this will allow the user to distinguish between node types. 

For this example I'll be using 3 bitmaps for the nodes. One for the root, and two for each child node - 1 for when the node is selected and 1 for when it is not. 

I'll be continuing with the application started in example 1. The full source for this example (including the bitmaps) is available from the "Source Download" page. 

Drop a TImageList component [image: image1.png]


onto the form. Double click the image list to add the 3 bitmaps. First the root, then closed node, and finally the open node bitmap. 


Your screen should look something like this 

Set the TTreeView's Images property to the name of the TImageList, which will be ImageList1 (unless you've renamed it, or have other ImageLists). 

Define constants to represent the index of the images in the list. This is useful as it allows you to move the images in the list at a later stage, without having to change every line of code that referances them. 

   const

      IMG_NODE_ROOT = 0;

      IMG_NODE_CLOSED = 1;

      IMG_NODE_OPEN = 2;

Here is the updated source for but_Add's onClick event. 



procedure TForm1.but_AddClick(Sender: TObject);

var

   sText : string;

begin

      {If nothing is selected}

   if(  tv_eg1.Selected = nil  ) then

   begin

         {Does a root node already exist?}

      if(  tv_eg1.Items.Count = 0  ) then

      begin

            {Add the root node}

         with tv_eg1.Items.AddFirst(  nil,  'Root'  ) do

         begin

            Selected := true;

               {Set the roots image index}
            ImageIndex := IMG_NODE_ROOT;

               {Set the roots selected index. The same image is uses

                  as for the ImageIndex}
            SelectedIndex := IMG_NODE_ROOT;

         end;

      end

      else begin

            {There is a root, so user must first select a node}

         MessageBeep(  -1  );

         ShowMessage(  'Select a parent node'  );

         Exit;

      end;

   end

   else begin

         {Get a name for the new node}

      InputQuery(  'New Node',  'Caption ?',  sText  );

         {Add the node as a child of the selected node}

      with tv_eg1.Items.AddChildFirst(  tv_eg1.Selected,  sText  ) do

      begin

            {Set the image used when the node is not selected}
         ImageIndex := IMG_NODE_CLOSED;

            {Image used when the node is selected}
         SelectedIndex := IMG_NODE_OPEN;

         MakeVisible;

      end;

   end;

end;




As you can see from the source the two properties we're interested in are ImageIndex and SelectedIndex. 

	ImageIndex
	Index of the image, in the TImageList, to display when the node is not selected. 

	SelectedIndex
	Index of the image, in the TImageList, to display when the node is selected. 


Giving ImageIndex and SelectedIndex the same value is entirely valid. It just means that the node's image does not change when it selected or unselected. 

Run example 2 and you'll see what a differance images make to an application.

TTreeNode.Level
Knowing at what level a node is at in the tree can often be very usefull. Getting this information is easy enough, just look at the TTreeNode's Level property. 

The pic shows that the root is at level 0, all nodes that are childern of the root are level 1 etc... 

Preventing Duplicates

A very common requirement for trees, is that all node names belonging to a parent have unique names. This is similar to the restriction on file names, all file names in a directory must be unique. However there is no built into restriction in TTreeViews, so this checking must be done manually. 

Here is a function to check if an name is already in use 

   function IsDuplicateName(  Node : TTreeNode;

                              sNewName : string;

                              bInclusive : boolean

                            ) : boolean;

   var

      TestNode : TTreeNode;

   begin

      if(  Node = nil  ) then

      begin

         Result := false;

         Exit;

      end;

         {Include this Node?}
      if(  bInclusive  ) then

         if(   CompareText(  Node.Text,  sNewName  ) = 0   ) then

         begin

            Result := true;

            Exit;

         end;

         {Test all previous siblings}
      TestNode := Node;

      repeat

            {Get next}
         TestNode := TestNode.GetPrevSibling;

         if(  TestNode <> nil  ) then

               {Is this a duplicate}
            if(   CompareText(  TestNode.Text,  sNewName  ) = 0   ) then

            begin

               Result := true;

               Exit;

            end;

      until (TestNode = nil);

         {Test all next siblings}
      TestNode := Node;

      repeat

            {Get next}
         TestNode := TestNode.GetNextSibling;

         if(  TestNode <> nil  ) then

               {Is this a duplicate}
            if(   CompareText(  TestNode.Text,  sNewName  ) = 0   ) then

            begin

               Result := true;

               Exit;

            end;

      until (TestNode = nil);

      Result := false;

   end;


This function checks for duplicate nodes using the following alorighm

1. If this is an inclusive search 

· Check if the node passed in param 1 name is sName 

2. Check all siblings prior to the node in param 1 

3. Check all siblings after the node in param 1 

Here is the updated source for but_Add's onClick event 



   procedure TForm1.but_AddClick(Sender: TObject);

   var

      sText : string;

   begin

         {If nothing is selected}

      if(  tv_eg1.Selected = nil  ) then

      begin

            {Does a root node already exist?}

         if(  tv_eg1.Items.Count = 0  ) then

         begin

               {Add the root node}

            with tv_eg1.Items.AddFirst(  nil,  'Root'  ) do

            begin

               Selected := true;

                  {Set the roots image index}

               ImageIndex := IMG_NODE_ROOT;

                  {Set the roots selected index. The same image is uses

                     as for the ImageIndex}

               SelectedIndex := IMG_NODE_ROOT;

            end;

         end

         else begin

               {There is a root, so user must first select a node}

            MessageBeep(  -1  );

            ShowMessage(  'Select a parent node'  );

            Exit;

         end;

      end

      else begin

            {Get a name for the new node}

         InputQuery(  'New Node',  'Caption ?',  sText  );

            {Check if this name is already in use}
         if(   IsDuplicateName(  tv_eg1.Selected.GetFirstChild,

                                 sText,

                                 true

                               )

            ) then

         begin

            MessageBeep(  -1  );

            ShowMessage(  'A node with this name already exists'  );

            Exit;

         end;

            {Add the node as a child of the selected node}

         with tv_eg1.Items.AddChildFirst(  tv_eg1.Selected,  sText  ) do

         begin

               {Set the image used when the node is not selected}

            ImageIndex := IMG_NODE_CLOSED;

               {Image used when the node is selected}

            SelectedIndex := IMG_NODE_OPEN;

            MakeVisible;

         end;

      end;

   end;

TTreeNode.Data

Each TreeNode has a .Data property. This property is useful, in that it can be used to associate additional data with a TTreeNode. You could for instance associate each TreeNode's data property with a Form, and display the associated form when the TreeNode is clicked, or whatever... 

TTreeNode.Data is an untyped pointer, which means it can point to anything. There are a few ways you can use this Data property. I'll demonstrate 3 of them. 



The first method is very simple, but can only be used when you want to keep an additional integer value with each TTreeNode. Here is a short example 

      {Make Node.Data contain the integer value 1234}
   Node.Data := pointer(1234);

      {Get the integer value}   

   iValue := integer(Node.Data);


All I'm doing here is treating the pointer as an integer value. It is extremely important that you don't try use the TTreeNode.Data as a real pointer in this case. You will undoubtedly crash your application...
In the example above the .Data property contains the integer value 1234. If you try use Node.Data as a pointer...... who knows? After all you have no idea what is at memory address 1234 ($4d2). So just don't do it! 



The second method uses classes, this is the method that most people will probably use. Its easy and flexible and there are no pointers to worry about (3rd method). 

The reason this method works so well is that Delphi conceals the fact that a class instance is actually a pointer. 

var

   MyClass : TMyClass

begin

   MyClass := TMyClass.Create;

   MyClass.Free;

end;

The MyClass variable is actually a pointer! But thanks to Delphi you never need to worry about that. So how does this help? Well look at this example 

   Node.Data := TMyClass.Create;

   TMyClass(Node.Data).Free;

The first line creates an instance of TMyClass and assigns it to the TTreeNode.Data property. The second line type-casts the pointer and frees it. The reason you need to type-cast is that the TTreeNode.Data property is an untyped pointer. Delphi has no idea that you want it to store a TMyClass instance. 

The following source is modified from example 1. Here each node has an associated TNodeData class. This class stores that time that the node was created as a string. This is a pretty useless example, but demonstrates how this method works. 

	Control
	Caption
	Name

	TButton
	Add
	but_Add

	TButton
	Remove
	but_Remove

	TTreeView
	 
	tv_eg1

	TEdit
	 
	ed_Time


type

  TNodeData = class

     sText : string;

  end;

   procedure TForm1.but_AddClick(Sender: TObject);

   var

      sText : string;

   begin

         {If nothing is selected}

      if(  tv_eg1.Selected = nil  ) then

      begin

            {Does a root node already exist?}

         if(  tv_eg1.Items.Count = 0  ) then

         begin

               {Add the root node}

            with tv_eg1.Items.AddFirst(  nil,  'Root'  ) do

            begin

               Selected := true;

                  {Create the data class}
               Data := TNodeData.Create;

                  {Set the nodes date time}
               TNodeData(Data).sText := FormatDateTime(  'hh:nn:ss',  now  );

            end;

         end

         else begin

               {There is a root, so user must first select a node}

            MessageBeep(  -1  );

            ShowMessage(  'Select a parent node'  );

            Exit;

         end;

      end

      else begin

            {Get a name for the new node}

         InputQuery(  'New Node',  'Caption ?',  sText  );

            {Add the node as a child of the selected node}

         with tv_eg1.Items.AddChild(  tv_eg1.Selected,  sText  ) do

         begin

               {Create the data class}
            Data := TNodeData.Create;

               {Set the nodes date time}
            TNodeData(Data).sText := FormatDateTime(  'hh:nn:ss',  now  );

            MakeVisible;

         end;

      end;

   end;

procedure TForm1.but_RemoveClick(Sender: TObject);

begin

      {Make sure somthing is selected, before trying to

        delete it}

   if(  tv_eg1.Selected = nil  ) then

   begin

      MessageBeep(  -1  );

      ShowMessage(  'Nothing selected'  );

      Exit;

   end;

      {don't allow user to delete the root node}

   if(  tv_eg1.Selected.Level = 0  ) then

   begin

      MessageBeep(  -1  );

      ShowMessage(  'Cant delete the root node'  );

      Exit;

   end;

      {Free the class}
   if(  tv_eg1.Selected.Data <> nil  ) then

      TNodeData(tv_eg1.Selected.Data).Free;

      {Delete the node}

   tv_eg1.Selected.Delete;

end;

There are a few important lines in the source above 

  TNodeData = class

     sText : string;

  end;

Define the class that is going to be associated with each TTreeNode. 

   Data := TNodeData.Create;   

Create the TNodeData class and assign it to the Data property 

   TNodeData(Data).sText := FormatDateTime(  'hh:nn:ss',  now  ); 

Set the TNodeData's creation time. 

   if(  tv_eg1.Selected.Data <> nil  ) then

      TNodeData(tv_eg1.Selected.Data).Free;

Free the memory used by the TNodeData class when it is no longer needed 

   procedure FreeAll_callback(  Node : TTreeNode;  pData : pointer  );

   begin

      if(  Node.Data <> nil  ) then

      begin

            {Free the memory used by the node}
         TNodeData(Node.Data).Free;

         Node.Data := nil;

      end;

   end;

   procedure TForm1.FormClose(Sender: TObject; var Action: TCloseAction);

   begin

         {If there are no nodes, exit}
      if(  tv_eg1.Items.Count <= 0  ) then

         Exit;

         {Call FreeAll_callback once for every node in the tree}
      EnumNodes(  tv_eg1.Items[ 0 ],

                  FreeAll_callback,

                  nil,

                  false

                );

   end;

It is important to remember to free all nodes when the application is closed. In the code above the EnumNodes procedure is used to enumerate all the nodes. See Pre-Ordal and Post-Ordal recursive tree traversals for info on this procedure. 



Finally method 3. This method is very similar to the above one, however instead of using a class I'll use a record. I'll just show the difference between this and method 2... 

     {The record pointed to by each Node's .data property}
  prNodeData = ^rNodeData;

  rNodeData = record

     sText : string;

  end;

Instead of a class declare a record. Also declare a pointer to the record. 

      {Allocate memory for the record}
   Data := New(  prNodeData  );

      {Set the nodes date time}
   prNodeData(Data)^.sText := FormatDateTime(  'hh:nn:ss',  now  );

First allocate memory for the record. Then set the sText field. Remember to dereferance ( ^ ) the pointer. 

      {Free the memory used by the record}
   if(  tv_eg1.Selected.Data <> nil  ) then

      Dispose(  prNodeData(tv_eg1.Selected.Data)  );

Free the memory used by the record. 

   procedure FreeAll_callback(  Node : TTreeNode;  pData : pointer  );

   begin

      if(  Node.Data <> nil  ) then

      begin

            {Free the memory used by the node}
         Dispose(  prNodeData(Node.Data)  );

         Node.Data := nil;

      end;

   end;

   procedure TForm1.FormClose(Sender: TObject; var Action: TCloseAction);

   begin

         {If there are no nodes, exit}
      if(  tv_eg1.Items.Count <= 0  ) then

         Exit;

         {Call FreeAll_callback once for every node in the tree}
      EnumNodes(  tv_eg1.Items[ 0 ],

                  FreeAll_callback,

                  nil,

                  false

                );

   end;

Again remember to free all memory used when the application is closed. (See Pre-Ordal and Post-Ordal recursive tree traversals)

BeginUpdate..EndUpdate

When adding more than 1 Node to a TTreeView its wise to call BeginUpdate and EndUpdate. BeginUpdate stops all TreeView redraws while nodes are being added to the TreeView. EndUpdate displayes all the updates that were made. 

It is suprising how much you can speed up your applications by using BeginUpdate..EndUpdate. Its so easy to do, it be great if all optimizations were this easy! 

   var

      iLoop : integer;

      sStart : string;

   begin

         {Word with an empty tree}
      TreeView.Items.Clear;

         {Start time}
      sStart := FormatDateTime(  'hh:nn:ss',  now  );

         {Prepare for update}
      TreeView.Items.BeginUpdate;

         {Add 1000 root items}
      for iLoop := 0 to 1000 do

         TreeView.Items.AddFirst(  nil,  IntToStr(  iLoop  )  );

         {Finished update}
      TreeView.Items.EndUpdate;

         {Show start + end times}
      ShowMessage(   sStart + #13#10 +

                     FormatDateTime(  'hh:nn:ss',  now  )

                  );

   end;


Try the above code with and without the BeginUpdate..EndUpdate calls. On my machine the results were quite astounding! With BeginUpdate..EndUpdate the function executed in less than 1 second. Without the BeginUpdate..EndUpdate calls it took about 30 seconds! 

One last thing the Win32 help says Tree nodes affected by the changes will have invalid Index values until EndUpdate is called. So dont use Node Indexes while in a BeginUpdate..EndUpdate "Block". 

ItemID - Finding Nodes

A common problem encountered when using TreeViews is finding a node.... Say you want to keep a list of the last few nodes the user clicked, how would you do that? 

Actually its relatively simple. Each node is uniquely identified by a ItemID property. The ItemID property is declared as a HTreeItem. HTreeItem is defined in CommCtrl.pas you must include this unit if you want to use HTreeItem. 

There are two steps here. 

Step 1: Get the Node's ItemID

   iValue := integer(Node.ItemID);

Here I get the node's ItemID and save it as an integer (type-cast is needed). 

Step 2: Get a Node from the ItemID

   Node := TreeView.Items.GetNode(  HTreeItem(iValue)  );

Again a type-cast is necessary, this time from an integer to a HTreeItem. 



Example App. 

	Control
	Caption
	Name

	Add
	but_Add

	TButton
	Go
	but_Go

	TEdit

This application is rather pointless, it has no real purpose. It should however show quite clearly how to use ItemIDs.A more practical use of ItemIDs will be demonstrated in the "Linking a TreeView and a ListView". 

This example is based on example 1. 

   procedure TForm1.tv_eg1Click(Sender: TObject);

   begin

         {If there is no selection then exit}
      if(  tv_eg1.Selected = nil  ) then

         Exit;

         {Get the node's ItemID,

            type-cast it to an integer,

            then convert to a string

            and add it to the TMemo}
      mem_LastNodes.Lines.Add(   IntToStr(  integer(tv_eg1.Selected.ItemID)  )    );

   end;

When a node is selected, first check that there is really a selection. Then get the selected Node's ItemID and display it in the TMemo. 

   procedure TForm1.but_GoClick(Sender: TObject);

   var

      iNodeID : integer;

      FoundNode : TTreeNode;

   begin

         {Is there text available?}
      if(  ed_GoTo.Text = ''  ) then

         Exit;

         {Get the ItemID the user entered (as an integer)}
      iNodeID := StrToIntDef(  ed_GoTo.Text,  0  );

      if(  iNodeID = 0  ) then

         Exit;

         {Now find the node.

            Type cast the integer to a HTreeItem.  To do this you

               must include  Commctrl.pas!

            Call GetNode}
      FoundNode := tv_eg1.Items.GetNode(  HTreeItem(iNodeID)  );

          {If a node was found,  select it}
      if(  FoundNode <> nil  ) then

         FoundNode.Selected := true;

   end;

A ItemID has been entered in the TEdit. Convert the string to a number, the number to a ItemID and then find the node and select it.

Pre-Order and Post-Order recursive tree traversals

Thats an exciting topic heading is'nt it ;-) 

A traversal is just "visiting" each node in the tree in a specific order. I'll discuss Pre-Order and Post-Order traversals here. Performing an operation on each node and an important task for many applications. What is also important is the order in which the nodes are visited. 

Here is example 8 

Clicking the traversal button with Pre-Order checked will, obviously, start a Pre-Order traversal. Without Pre-Order checked a Post-Order traversal is performed. 

Here are the results of these two traversals with the tree structure show in the picture above. 

	Traversals Order
	Node order

	Pre-Order
	Root -> 1 -> 2 -> 4 -> 6 -> 7 -> 5 -> 8 -> 9 -> 10 -> 3

	Post-Order
	7 -> 6 -> 10 -> 9 -> 8 -> 5 -> 4 -> 3 -> 2 -> 1 -> Root 


The Pre-Order traversal starts at the root, processes the root. Then moves one node left and processes that node. If there is not a left node it tries to move right. If there is no right node it moves up one and tries moves right. Look at the example and follow the path the traversal takes, its very simple once you see it. 

A Post-Order traversal is very similar in code to the Pre-Order traversal. The difference is that the function first moves as far left as possible and then only processes the node. Again follow the path taken by the traversal in the example. 



   procedure EnumNodes(  Node : TTreeNode;

                         Enum : EnumNodesProc;

                         pData : pointer;

                         bPre : boolean

                       );

   begin

      //////////////////////////////////////////////////////////////////////

      //  Enter Function              <------+  <---+

      //                                     |      |

      //    Process this node  (if Pre)      |      |

      //                                     |      |

      //    If this node has children        |      |

      //       Call Enum with  first child---+      |

      //                                            |

      //    If this node has siblings               |

      //       Call Enum with  next sibling---------+

      //

      //

      //    Process this node  (if not Pre)

      //////////////////////////////////////////////////////////////////////

      // EnumNodesProc = procedure(  Node : TTreeNode;  pData : pointer  );

      ///////////////////////////////////////////////////////////////////////
      if(  Node = nil  ) then

         Exit;

         {Preorder}
      if(  bPre  ) then

         Enum(  Node,  pData  );

      if(  Node.GetFirstChild <> nil  ) then

         EnumNodes(  Node.GetFirstChild,  Enum,  pData,  bPre  );

      if(  Node.GetNextSibling <> nil  ) then

         EnumNodes(  Node.GetNextSibling,  Enum,  pData,  bPre  );

         {Postorder}
      if(  not bPre  ) then

         Enum(  Node,  pData  );

   end;

As you can see the only differance between Pre-Order and Post-Order is when the node is proccessed. With recursive functions this makes a big differance. 

This funtion uses a call back procedure that is defined 

   EnumNodesProc = procedure(  Node : TTreeNode;  pData : pointer  );

The Node parameter is the node to be processed, the pData parameter is a user defined pointer.

Editing TreeNode.Text

   procedure TForm1.tv_eg1Editing(Sender: TObject; Node: TTreeNode;  var AllowEdit: Boolean);

   begin

         {All but root may be edited}

      AllowEdit := (Node <> tv_eg1.Items[ 0 ]);

   end;

The OnEditing event is used to allow or disallow editing of a nodes text. In this example the only restriction is that the root node may not be renamed. (This event is called before the node is edited) 



   procedure TForm1.tv_eg1Edited(Sender: TObject; Node: TTreeNode;  var S: String);

   begin

      mem_History.Lines.Insert(  0,  Node.Text + ' --> renamed --> ' + S  );

   end;

The OnEdited is called after an edit has taken place. Node.Text still has the Node's original name. The variable S contains the new name. You can change S back to the original name, or to any other value as you see fit. 

Here all renames are logged to the TMemo. 



   procedure TForm1.tv_eg1KeyUp(Sender: TObject; var Key: Word;  Shift: TShiftState);

   begin

      case Key of

         VK_F2 :

         begin

            if(  tv_eg1.Selected <> nil  ) then

               tv_eg1.Selected.EditText;

         end;

         VK_DELETE :

         begin

            if(  tv_eg1.Selected <> nil  ) then

               if(  not tv_eg1.IsEditing  ) then

                  but_Remove.Click;

         end;

      end;

   end;

The OnKeyUp event is used here to allow the user to use the F2 and Delete keys to make working with the TreeView easier. 

	F2
	If there is a selected node then go into edit mode (tv_eg1.Selected.EditText) 

	Delete
	1. If there is a selected node 

2. If not in edit mode 

3. Delete the node 


The "if not in edit mode" test may seem unnecessary, but its not. When the user is in edit mode and pressed Delete, then the character after the insertion pointer must be deleted. If this check where not there, then pressing Delete while in edit mode would delete the selected node. 


A more complex example

The project started in example one, is not complex enough to demonstrate some of the mode advanced TTreeView related topics. This example is slightly more complex, but is not much more difficult to understand. 

In the previous examples there have been two node types - the root and "normal" nodes. Here there a 3 node types - the root, file nodes and folder nodes. 

As with example 1 I'm going to use constants to index the node images. 

   const

      IMG_NODE_ROOT = 0;

      IMG_NODE_FILE_CLOSED = 1;

      IMG_NODE_FILE_OPEN = 2;

      IMG_NODE_FOLDER_CLOSED = 3;

      IMG_NODE_FOLDER_OPEN = 4;


Here are the "rules" for the nodes for this example

· There may only be one root node 

· The root node may contain any number of files or folders 

· A folder may contain other folders and/or files 

· A file may contain nothing (a "leaf" node) 


In example 2 we saw how to give node's images. Here there are more types of nodes, but otherwise its exactly the same. What is different is that I'm going to be using a node's ImageIndex to identify it. 

I've also defined an enum for each type of node, and one for node's who's type can not be determined

      {Enum used for easily identifying nodes}
   eNodeType = (ntUnknown,  ntRoot,  ntFile,  ntFolder);


Here is the function that determines what type of node we're working with 

   //////////////////////////////////////////////////////////////

   // Returns one of the eNodeType values to indicate what type

   //  of node param 1 is.

   //////////////////////////////////////////////////////////////
   function TForm1.GetNodeType(  Node : TTreeNode  ) : eNodeType;

   begin

      if(  Node = nil  ) then

      begin

         Result := ntUnknown;

         Exit;

      end;

         {Determine what type of node this is by looking at the

           node's ImageIndex}
      case Node.ImageIndex of

         IMG_NODE_ROOT : Result := ntRoot;

         IMG_NODE_FILE_CLOSED : Result := ntFile;

         IMG_NODE_FOLDER_CLOSED : Result := ntFolder;

      else

            {Node should be one of the above...}
         Result := ntUnknown;

      end;

   end;


Example 5's form looks like this

	Control
	Caption
	Name

	TTreeView
	 
	tv_eg1

	TButton
	Add Folder
	but_AddFolder

	TButton
	Add File
	but_AddFile

	TButton
	Remove
	but_Remove

	TImageList
	 
	ImageList1



As you can see the buttons are labeled "but_AddFolder" and "but_AddFile". So when, you might ask, do we add the root node? In the OnCreate event... 

   procedure TForm1.FormCreate(Sender: TObject);

   begin

         {Add the root node}
      AddRootNode;

   end;

   ///////////////////////////////////

   // Adds the root to an empty tree

   ///////////////////////////////////
   procedure TForm1.AddRootNode;

   begin

         {If the tree is empty}
      if(  tv_eg5.Items.Count = 0  ) then

      begin

            {Add the root node}
         with tv_eg5.Items.AddFirst(  nil,  'Root'  ) do

         begin

            Selected := true;

               {Set the roots image index}
            ImageIndex := IMG_NODE_ROOT;

               {Set the roots selected index. The same image is uses

                  as for the ImageIndex}
            SelectedIndex := IMG_NODE_ROOT;

         end;

      end

   end;


Before adding nodes lets create a function that will check if a node may be added to any given parent. ie A function that checks the node's type and position against the "rules" 

   ///////////////////////////////////////////////////////

   // Check if a new node of type NewNodeType may be

   //   created as a child off ParentNode

   ///////////////////////////////////////////////////////
   function TForm1.IsNodeAllowed(  ParentNode : TTreeNode;

                                   NewNodesType : eNodeType

                                 ) : boolean;

   begin

      case GetNodeType(  ParentNode  ) of

         ntRoot :

         begin

               {A root may contain any type of node}
            Result := true;

         end;

         ntFolder :

         begin

               {Folder may contain any type of node}
            Result := true;

         end;

         ntFile :

         begin;

               {Files may have no sub-items}
            Result := false;

         end;

      else

            {Unknown node type, dont allow any operations}
         Result := false;

      end;

   end;


And finally the procedure to add a new node 

   ///////////////////////////////////////////////////

   // Procedure used to add a file / folder node

   ///////////////////////////////////////////////////
   procedure TForm1.AddNode(  NodeType : eNodeType  );

   var

      sText : string;

   begin

         {If nothing is selected}
      if(  tv_eg5.Selected = nil  ) then

      begin

            {There is a root, so user must first select a node}
         MessageBeep(  -1  );

         ShowMessage(  'Select parent node'  );

         Exit;

      end

      else begin

            {Get a name for the new node}
         if(   not InputQuery(  'New Node',  'Caption ?',  sText  )   ) then

            Exit;

            {Check if this name is already in use}
         if(   IsDuplicateName(  tv_eg5.Selected.GetFirstChild,

                                 sText,

                                 true

                               )

            ) then

         begin

            MessageBeep(  -1  );

            ShowMessage(  'A node with this name already exists'  );

            Exit;

         end;

            {Check if adding this type of node is allowed}
         if(   not IsNodeAllowed(  tv_eg5.Selected,  NodeType  )   ) then

         begin

            MessageBeep(  -1  );

            ShowMessage(  'Cant creat this type of node here'  );

            Exit;

         end;

            {Add the node}
         with tv_eg5.Items.AddChildFirst(  tv_eg5.Selected,  sText  ) do

         begin

            case NodeType of

               ntFolder :

               begin

                     {Set the image used when the node is not selected}
                  ImageIndex := IMG_NODE_FOLDER_CLOSED;

                     {Image used when the node is selected}
                  SelectedIndex := IMG_NODE_FOLDER_OPEN;

                  MakeVisible;

               end;

               ntFile :

               begin

                     {Set the image used when the node is not selected}
                  ImageIndex := IMG_NODE_FILE_CLOSED;

                     {Image used when the node is selected}
                  SelectedIndex := IMG_NODE_FILE_OPEN;

                  MakeVisible;

               end;

            else

                  {Trying to add a node that is not a file or a folder,

                    this is not allowed. So remove the node that was

                    just created.}
               Delete;

            end;

         end;

      end;

   end;


As you can see this procedure is quite differant from the one in example 1. However the underlying ideas are exactly the same. There is additional checking, and "rule" enforcement. 

Well thats example 5. The next few examples will be based on this one, so understanding how this example works is well worth the time.

Tree Sorting

There are two ways to sort a tree. The first is extremely easy, just use the TTreeView.AlphaSort method. The tree will, obviously enough, be sorted alphabetically. 

This type of simple sorting is more suitable for "simple" trees, like example 1 - 4. It is not good enough for more complex trees though. 

Here is an example of an Alpha sorted tree 

The problem here is that folder and file nodes are not grouped together. Optimally the same tree, once sorted, should look like this 

[image: image11.png][_[OIx]

Add Folder
AddFie

Remave






In this tree files are always displayed before folders. The more complex the tree the more useful "grouping" becomes. TreeView's support user defined sorting, the method to call is TTreeView.CustomSort. 

The CustomSort function takes 2 parameters

	SortProc : TTVCompare
	SortProc is a callback function. This function is called by the TTreeView and compares pairs of nodes. 

	Data : LongInt
	You can pass any value to the callback function using this parameter. If you want to pass a pointer then just type cast it to an integer.
eg CustomSort( ..., integer(@Somthing) );


The callback function that is passed as param 1 is defined as follows

function(  Node1,  Node2 : TTreeNode;  Data : integer  ) : integer; stdcall;

What you have to do is compare Node1 and Node2, then return 

	Positive integer
	Node2 displayed before Node1

	0
	Nodes are equivalent

	Negative integer
	Node1 displayed before Node2




The example callback function 

   /////////////////////////////////////

   // Custom sort callback function

   /////////////////////////////////////
   function MyCustomSortProc(  Node1,  Node2 : TTreeNode;  Data : integer  ) : integer; stdcall;

         {Inline function returns true if Node is a File}
      function IsAFile(  Node : TTreeNode  ) : boolean;

      begin

         Result := ((Node.ImageIndex = IMG_NODE_FILE_CLOSED) or

                    (Node.ImageIndex = IMG_NODE_FILE_OPEN));

      end;

         {Inline function returns true if Node is a Folder}
      function IsAFolder(  Node : TTreeNode  ) : boolean;

      begin

         Result := ((Node.ImageIndex = IMG_NODE_FOLDER_CLOSED) or

                    (Node.ImageIndex = IMG_NODE_FOLDER_OPEN));

      end;

   begin

         {Files before folders}
      if(   IsAFile(  Node1  ) and IsAFolder(  Node2  )   ) then

      begin

         Result := -1;

         Exit;

      end;

         {Folder after file}
      if(   IsAFolder(  Node1  ) and IsAFile(  Node2  )   ) then

      begin

         Result := 1;

         Exit;

      end;

         {Nodes are of the same type, so do a normal alpha sort}
      Result := AnsiStrIComp(  PChar(Node1.Text), PChar(Node2.Text)  );

   end;

Calling the sort routine 

   tv_eg5.CustomSort(  @MyCustomSortProc,  0  );


I've added the CustomSort call to the AddNode procedure. This means that every time a node is added the tree is sorted. For this example that is exactly what must happen. If you were loading a tree then you'd only want to sort it once the entire load operation had been completed.

Linking a TreeView and a ListView

Here is what the finished application will look like. 

	Control
	Caption
	Name

	TButton
	Add Folder
	but_AddFolder

	TButton
	Add File
	but_AddFile

	TButton
	Remove
	but_Remove

	TTreeView
	 
	tv_eg5

	TListView
	 
	lv_Link


· Selecting a node in the TreeView will... 

· If the node has a parent a "To Parent" icon will be added to the ListView 

· Every child of the selected node will be displayed in the List View 

· Double-clicking a item in the ListView will select the corresponding item in the TreeView. 

· Renaming an item in the ListView, will rename the corresponding item in the TreeView. 


Looking at the above requirements, the most difficult is probably selecting a TreeView node when a ListView item is double clicked. This is quite easily solved by using the Node's ItemID property (see example 7). 

As explained in example 5 I'll be using the ListView's data property as if it was an integer. In a real application you might want to store the ItemID value as a member of a class / record that the .Data property points to (example 5). 

The way to link a ListItem to a TreeNode is to store the TreeNode's ItemID in the ListItem's data property. eg 

   ListView.Data := TreeNode.ItemID;

Now when the ListItem is double clicked you get the node with the given TreeNode eg 

   LinkNode := TreeView.Items.GetNode(  HTreeItem(ListView.Data)  );

And then select the TreeNode 

   LinkNode.Selected := true;

As you can see its very easy. Once you get the TreeNode that the ListItem is linked to you can do whatever you like with it. Rename, delete, move etc. 


Before moving on to the source, some notes on the ListView.

· ViewStyle is set to vsSmallIcon
· SmallImages is assigned the same ImageList as used by the TreeView

· A "To Parent"[image: image13.png]


 icon has been added to the ImageList as item 5 (IMG_TO_PARENT). 



The source is based on example 10, so there are folder, file and root nodes. 

Here is the procedure that displays the TreeNodes in the ListView. 

   procedure TForm1.UpdateListView;

   var

      ChildNode : TTreeNode;

   begin

      lv_Link.Items.Clear;

         {If nothing is selected}
      if(  tv_eg5.Selected = nil  ) then

         Exit;

         {Does this node have a parent?}
      if(  tv_eg5.Selected.Parent <> nil  ) then

      begin

           {Add the "up to parent" node}
         with lv_Link.Items.Add do

         begin

            Caption := 'To Parent';

               {Set the image}
            ImageIndex := IMG_TO_PARENT;

               {Node that ListItem is linking to (ie Parent Node)}
            Data := tv_eg5.Selected.Parent.ItemId;

         end;

      end;

      ////////////////////////////////////////////////

      // Now get all the selected node's child nodes

      ////////////////////////////////////////////////
      lv_Link.Items.BeginUpdate;

         {Get first child}
      ChildNode := tv_eg5.Selected.GetFirstChild;

      while(  ChildNode <> nil  ) do

      begin

           {Add the "up to parent" node}
         with lv_Link.Items.Add do

         begin

            Caption := ChildNode.Text;

               {Set the image}
            ImageIndex := ChildNode.ImageIndex;

               {Node that ListItem is linking to}
            Data := ChildNode.ItemId;

         end;

            {Next child node}
         ChildNode := tv_eg5.Selected.GetNextChild(  ChildNode  );

      end;

      lv_Link.Items.EndUpdate;   

   end;

Nothing to complex here.. 

1. Clear old items from the ListView

2. If the node has a parent add the To Parent icon

3. Add every child TreeNode as a ListItem, with the same ItemIndex and save the Node's ItemID in the ListItem's Data property 


This procedure must be called whenever a change is made to the TreeView so add it to the TreeView's OnChange event 

   procedure TForm1.tv_eg5Change(Sender: TObject; Node: TTreeNode);

   begin

      UpdateListView;

   end;

As well as to AddNode and RemoveClick 




Next the ListView's OnClick event 

   procedure TForm1.lv_LinkDblClick(Sender: TObject);

   var

      LinkNode : TTreeNode;

   begin

         {Make sure there is a selection}
      if(  lv_Link.Selected = nil  ) then

         Exit;

         {Find the node in the TreeView that corresponds to

           this ListView Item}
      LinkNode := tv_eg5.Items.GetNode(  HTreeItem(lv_Link.Selected.Data)  );

         {If the node was found, select it}
      if(  LinkNode <> nil  ) then

         LinkNode.Selected := true;

   end;

This too is very simple. 

1. Check that there is a selection 

2. Try locate the corresponding TreeNode 

3. If the node was found select it 




Finally the ListView's OnEdited and OnEditing events 

   procedure TForm1.lv_LinkEditing(Sender: TObject; Item: TListItem;  var AllowEdit: Boolean);

   begin

         {Can edit all but the "To Parent" item}

      AllowEdit := (Item.ImageIndex <> IMG_TO_PARENT);

   end;

Allow any Item except for the "To Parent" item to be renames 

   procedure TForm1.lv_LinkEdited(Sender: TObject; Item: TListItem;  var S: String);

   var

      LinkNode : TTreeNode;

   begin

         {Find the node that is being renamed}
      LinkNode := tv_eg5.Items.GetNode(  HTreeItem(Item.Data)  );

         {If the node was found}
      if(  LinkNode <> nil  ) then

      begin

            {Does another node already have this name?}
         if(   IsDuplicateName(  LinkNode,  s,  false  )   ) then

         begin

            MessageBeep(  -1  );

            ShowMessage(  'Duplicate Name!'  );

               {Revert to original name}
            S := Item.Caption;

         end

         else begin

               {Name is valid so allow rename}
            LinkNode.Text := s;

         end;

      end

      else begin

         {For some reason the link was not found, give ListIem its

           old name}
         S := Item.Caption;

      end;

   end;

Same as a "normal" OnEdited event, except for the Node linking. See example 9 for more information on OnEditing and OnEdited events. 



Thats all there is to it. Naturally there is a lot more that can be done. For instance allowing nodes to be added or deleted in the ListView. One way to do these kind of things is to have a procedure that is called by both the ListView and TreeView events. For the TreeView you'd call DoSomthing( TreeView.Selected ) for the ListView DoSomthing( LinkNode ), where LinkNode is found as above.

Drag and Drop

First a clarification, Drag and Drop here refers to drag and drop within the TreeView, or from the TreeView to another control within the same application. This tutorial does not cover drag and drop between applications, eg to/from Explorer. That has more to do with OLE than it does TreeViews. 

This example is based on example 10 and there are thus file and folder nodes as well as a root node. Take a look at the "rules" for this TreeView (example 10). These rules must be kept when nodes are dragged. 

The OnDragOver event is called when something (on this case a TreeNode) is dragged over a control (a TTreeView). You must indicate if the dragged item may be dropped at the current position. 

   procedure TForm1.tv_eg5DragOver(Sender, Source: TObject; X, Y: Integer;  State: TDragState; var Accept: Boolean);

   begin

   /////////////////////////////////////////

   // Decide if drag-drop is to be allowed

   /////////////////////////////////////////
      Accept := false;

         {Only accept drag and drop from a TTreeView}
      if(  Sender is TTreeView  ) then

            {Only accept from self}
         if(  TTreeView(Sender) = tv_eg5  ) then

            Accept := true;

   end;

1. Assume fail 

2. If the source is a TreeView 

3. If dragging internally ie tv_eg5 to ev_eg5 

4. Allow drag-drop 



The OnDragDrop event is called once an item is dropped on the TTreeView. This is a rather long function so a outline of the procedure will help clarify what exactly it does. 

1. Get the target node (the node that the item was dropped on) 

2. Get an alias for the source node. This makes the source easier to read 

3. Make sure the Target is a valid node 

4. Can the target node accept the source node 

1. Cant drop onto self, or drop onto immediate parent 

2. May not drag the root 

3. Cant drop a parent onto a child 

4. May not drop if an item with the same names as the source already exists 

5. Check the rules - IsNodeAllowed 

6. Copy the node 

7. Delete old node (Copy + Delete = move) 

8. Display node in its new position 

   procedure TForm1.tv_eg5DragDrop(Sender, Source: TObject; X, Y: Integer);

   var

      TargetNode : TTreeNode;

      SourceNode : TTreeNode;

   begin

   /////////////////////////////////////////

   // Somthing has just been droped

   /////////////////////////////////////////
      with tv_eg5 do

      begin

            {Get the node the item was dropped on}
         TargetNode := GetNodeAt(  X,  Y  );

            {Just to make things a bit easier}
         SourceNode := Selected;

            {Make sure somthing was droped onto}
         if(  TargetNode = nil  ) then

         begin

            EndDrag(  false  );

            Exit;

         end;

            {Dropping onto self or onto parent?}
         if(  (TargetNode = Selected) or

              (TargetNode = Selected.Parent)

           ) then

         begin

            MessageBeep(  MB_ICONEXCLAMATION  );

            ShowMessage(  'Destination node is the same as the source node'  );

            EndDrag(  false  );

            Exit;

         end;

            {No drag-drop of the root allowed}
         if(  SourceNode.Level = 0  ) then

         begin

            MessageBeep(  MB_ICONEXCLAMATION  );

            ShowMessage(  'Cant drag/drop the root'  );

            EndDrag(  false  );

            Exit;

         end;

            {Can't drop a parent onto a child}
         if(   IsAParentNode(  Selected,  TargetNode  )   ) then

         begin

            MessageBeep(  MB_ICONEXCLAMATION  );

            ShowMessage(  'Cant drop parent onto child'  );

            EndDrag(  false  );

            Exit;

         end;

            {Does a node with this name exists as a child of TargetNde}
         if(   IsDuplicateName(  TargetNode.GetFirstChild,  SourceNode.Text,  true  )   ) then

         begin

            MessageBeep(  MB_ICONEXCLAMATION  );

            ShowMessage(  'A node with this name already exists'  );

            EndDrag(  false  );

            Exit;

         end;

         //////////////////////////////////////////////////////////////

         // Nothing differant up to here.  Just the normal drag and

         //   drop checking.  Now the code to make sure that enforce

         //   "the rules".  Eg books may contain no sub-nodes

         //////////////////////////////////////////////////////////////
            {Use the IsNodeAllowed function to test if the node

              may be dropped here}
         if(  not IsNodeAllowed(   TargetNode,

                                   GetNodeType(  SourceNode  )

                                 )

            ) then

         begin

            MessageBeep(  -1  );

            ShowMessage(  'You cant drop this type of node here!'  );

            EndDrag(  false  );

            Exit;

         end;

            {Drag drop was valid so move the nodes}
         MoveTreeNode(  tv_eg5,  SourceNode,  TargetNode  );

            {Delete the old node}
         SourceNode.Delete;

            {Show the nodes that were just moved}
         TargetNode.Expand(  true  );

      end;

   end;



Drag and Drop between a TreeView and its linked ListView (eg 12) is a bit more difficult. Although not that much more...
Use a function that works similarly to the OnDragDrop function above, but TargetNode and SourceNode are passed as parameters 

Then...

	TreeView to ListView
	Get the ListItem the TreeNode was dropped onto. Get the TreeNode that this ListItem is linked to and use it as the TargetNode. 

	ListView.TreeView
	Get dragged TreeItem, get linked TreeNode use this as SourceNode 

	ListView to ListView
	Get source and target ListItems, get linked Source and Target TreeNodes. Call function 


I've not actually implemented this, but with the examples in this tutorial you should be able to get this working without too much trouble.

Saving To a File

The Delphi supplied SaveToFile and LoadFromFile methods, provide a very simple way to save a TreeView to / load it from a file. The problem with these methods is that they only save/load the TreeView structure. 

You can not save the ImageIndex, SelectedIndex, Data or any other TTreeNote property. For this reason many people will never use these methods, and instead will look for something more advanced. 

I personally use OLE Structured Storage to solve this problem. Structured Storage files, or DocFiles, support hierarchical data storage and thus are perfect for saving a TTreeView to disk. For an introduction to OLE Structured Storage, take a look at my DocFile Tutorial. Specifically the "Saving a TreeView to a DocFile" Example

